

```

1  ^ANODEN
10 VIDEO3:CLS:CLV
20 Y=5:LET[0],0,3,0,1,16,5,0,12,0,0,16,5
30 PLAY"TG00G2FEDRE2.":WAIT50
100 L=BTN(30):R=BTN(31):IFBTN(32)*PF=1:BEEP9*!X
110 LCX,Y:Y=?CHR$(0):IFLORRY=Y-(L*(Y>3)-R*(Y<7))*2ELSEY=5
120 X=X+F*2:IFX=16F=0:I=(C+10)%60:P=0:X=15
130 IFI=CP=1:X=0
140 LCX,Y:IF(X=[4])*(Y=[5])W=0:GSB410ELSEIF(X=[10])*(Y=[11])W=6:GS
B410ELSE?CHR$(45*P);
200 IF[3+W]*C%2GOTO260
210 LC[4+W],[5+W]:?CHR$(0):IF![4+W]BEEP80,6:D=D+1:GSB400:IF[5+W]=
5D=3
220 LC5,9:?"D: ";D;" /3 ";:IFD=3GOTO500
230 IF![W]*[2+W][W]=1:[5+W]=3+RND(3)*2
240 A=[5+W]:IF!RND(5)B=RND(3)-1:[5+W]=A+((A<7)-(A>3)+(A=5)*B)*2
250 [4+W]=[4+W]-[W]*2
260 A=[4+W]:LCA,[5+W]:IF!(SCR(A,[5+W])=45)?CHR$(241*[2+W]*!(A=1
6)):GOTO280
270 GSB410
280 LC5,1:?"S: ";S:IF[1+W]=C[2+W]=1:[3+W]=RND(2)
290 W=(W+6)%12:IFWGOTO200
300 C=(C+1)%60:BEEP90*([2]+[8]):WAIT9:GOTO100
400 [W]=0:[1+W]=(C+RND(9)+9)%60:[2+W]=0:[4+W]=16:RTN
410 BEEP19,9:?"*":WAIT5:I=(C+1+!F*9)%60:F=0:P=0:S=S+X/2:X=15:GSB4
00:CHR$(28,0):RTN
500 WAIT6:BEEP20:LC3,5:?"GAME OVER":CLK
510 IFINKEY()=10RUNELSECONT

```

■ 変数 ■

A、B：汎用（一時指定）
C：処理ループカウンタ（0～59をループ）
D：ダメージ数カウンタ（0～5）
F：自弾発射中フラグ（0or1）
I：自弾発射休止リミット値（撃ち漏らし時 C+10）
L：上ボタンフラグ（0or1）
P：自弾発射パーミッションフラグ（0or1）
R：下ボタンフラグ（0or1）
S：スコア
W：敵弾の配列変数処理に加算する係数（敵機0号処理時0、敵機1号処理時6）
X、Y：自弾の座標

敵機0号、1号

[0]、[6]：敵機発射中フラグ（0or1）
[1]、[7]：敵機発射休止リミット値（C+9～17）
[2]、[8]：敵機発進パーミッションフラグ（0or1）
[3]、[9]：敵機のスピード（0:速い、1:遅い）
[4][5]、[10][11]：敵機の座標

■ 処理の内訳 ■

10～30行：初期設定
100～140行：自弾処理
200～300行：敵機処理
400行：敵機変数リセット処理
410行：撃破処理
500～510行：ゲームオーバー処理

★ プログラムはIchigoJam BASIC 1.4系準拠でバランスをとっております。

1.5系でも動きますが動作が全体的に速くなるので、300行のWAIT値を11か12くらいに上げて調整してください。